
Espressioni Aritmetiche 
La scrittura e il calcolo di espressioni aritmetiche sono problemi, a prima vista semplici, che 
nascondono insospettate complicazioni e si prestano ad una interessante trattazione. 
Prendiamo come esempio il calcolo della espressione aritmetica rappresentata dall’albero: 

 
Nella notazione algebrica comunemente usata l'espressione si scrive:  
 (2+3)*(5+7) 
il cui valore è 60. Eliminare le parentesi porterebbe all'espressione:  
 2+3*5+7 
che comunemente viene valutata 24.  
Tale valutazione deriva dall’avere attribuita precedenza all’operatore *. Senza tale precedenza 
l'espressione sarebbe ambigua. Nessuna ambiguità invece si introduce omettendo l'operatore 
*:  
 (2+3)(5+7) 
e quest'ultima è la notazione più usata in algebra. Un ulteriore coppia di parentesi sarebbe 
ridondante ma renderebbe più omogenea la scrittura:  
 ((2+3)*(5+7)) 



 
Le notazioni polacche 
Coloro che hanno fatto esperienza di calcolo sulle calcolatrici Hewlett-Packard conosceranno 
già le cosiddette notazioni polacche,  
• diretta:  * + 2 3 + 5 7 
• inversa:  2 3 + 5 7 + * 
 
Le notazioni polacche, seppur prive di parentesi, non sono ambigue purché siano chiaramente 
delimitate le costanti numeriche (qui si è usato il blank come separatore altrimenti la sequenza 
2 seguito da 3 poteva essere confusa col valore 23).  
 
Infine, volendo evitare il cosiddetto zucchero sintattico, trattando, cioè, la somma e il prodotto 
come qualunque altra funzione di due variabili, si ha la notazione prefissa:  

prodotto(somma(2,3), somma(5,7)) 
Nel seguito ci limitiamo per semplicità a trattare le operazioni di addizione e moltiplicazione 
tra numeri di una cifra. Presentiamo dapprima per ciascuna forma la grammatica generativa.  



Le grammatiche delle varie notazioni 
 
• La notazione infissa ha la grammatica: 
    <numero>  ::=  0  |  1  |  2  |  3  |  4  |  5  |  6  |  7  |  8  |  9  
    <espr>  ::=  ( <espr> + <espr> )  |  ( <espr> * <espr> )  |  <numero>  
Questa grammatica prevede le parentesi ad ogni livello.  
 
• Più complessa è la specifica di una grammatica non ambigua che, assegnando la 
precedenza all'operatore di moltiplicazione, permetta di eliminare le parentesi superflue: 
    <numero>  ::=  0  |  1  |  2  |  3  |  4  |  5  |  6  |  7  |  8  |  9  
    <espr>  ::=  <espr> + <term>  |  <term>  
    <term>  ::=  <term> * <factor>  |  <factor>  
    <factor>  ::=  <numero>  |  ( <espr> )  
Questa grammatica non sarà più trattata nel seguito. 



• La notazione polacca diretta ha la grammatica: 
    <numero>  ::=  0  |  1  |  2  |  3  |  4  |  5  |  6  |  7  |  8  |  9  
    <espr>  ::=  + <espr> <espr>  |  * <espr> <espr>  |  <numero>  
 
• Molto simile è la notazione prefissa: 
 
    <numero>  ::=  0  |  1  |  2  |  3  |  4  |  5  |  6  |  7  |  8  |  9  
    <espr>  ::=  somma( <espr> , <espr> )  |  prod( <espr> , <espr> )  |   <numero> 
 
• La notazione polacca inversa ha la grammatica: 
    <numero>  ::=  0  |  1  |  2  |  3  |  4  |  5  |  6  |  7  |  8  |  9  
    <espr>  ::=  <espr> <espr> +  |  <espr> <espr> *  |  <numero> 
 


